ABSTRACT

As a Base SAS® programmer, you spend your day manipulating data and creating reports. You know there is a procedure that can give you what you want. As a matter of fact, there is probably more than one procedure to accomplish the task. Which one should you use? How do you remember which procedure is best for which task?

This paper is all about the Base procedures. It explores the strengths of the commonly used, nongraphing procedures. It discusses the challenges of using each procedure and compares it to other procedures that accomplish similar tasks. The first section of the paper looks at utility procedures that gather and structure data: APPEND, COMPARE, CONTENTS, DATASETS, FORMAT, SORT, SQL, and TRANSPOSE. The next section discusses the Base SAS procedures that work with statistics: FREQ, MEANS/SUMMARY, and UNIVARIATE. The final section provides information about reporting procedures: PRINT, REPORT, and TABULATE.

INTRODUCTION

The beauty of SAS is that usually there is more than one method to get what you need. That is a good thing! It can also be a challenge—you might not know if there is a more efficient method or a method that requires a lot less code.

This paper provides brief descriptions of the most frequently used Base SAS procedures. More importantly, it provides ideas for the best way to use the procedure and a comparison of when another procedure or technique might be better to use. Not every functionality of every procedure can be listed here. Nor can every comparison be made. Hopefully this paper will provide ideas on how to accomplish the tasks you need to accomplish, and it might encourage you to try multiple methods for accomplishing those tasks.

Please note that, due to space considerations, this paper focuses on using these procedures for SAS®9 data sets (and not for database tables).

UTILITY PROCEDURES

For the purposes of this paper, the utility procedures are those that enable you to get information about the data or to complete pre-analysis tasks such as restructuring the data.

APPEND PROCEDURE

Description: PROC APPEND is designed to add the observations in one data set (DATA= data set) to the bottom of another data set (BASE= data set). This process is also known as concatenating data sets.

Challenges:

- The data structure of the two data sets must match. The variable names, lengths, and types must be the same in the two data sets. By default, if the structures do not match, PROC APPEND will not append the observations. The FORCE option will force the append process to occur if the structure is different. However, in this case, variables in the DATA= data set that are not in the BASE= data set are not in the resulting data set.
- You cannot make changes to the observations in the BASE= data set or to the observations being added. The observations are simply appended.

Best Use Case: PROC APPEND has a very narrow focus—to add observations from one data set to another. It does not open the data set being added to. Therefore, it uses a limited amount of I/O. The
The ideal circumstance for using PROC APPEND is when you have routine or cyclical data and you need to add the latest data to the historical data.

**Example Code:**
```
data work.prdsal2;
   set sashelp.prdsal2;
run;

data sale2000;
   length country $10 state $22 county $20 prodtype product $10;
   input country $ state $ county actual predict prodtype $ product $ year quarter monyr monyy.;
   month=month(monyr);
datalines;
U.S.A California . 1702.50 40.00 FURNITURE BED 2000 1 JAN00
run;
proc append base=prdsal2 data=sale2000;
run;
```

**Comparisons:** A DATA step that uses the SET statement can accomplish the same task as PROC APPEND. With PROC APPEND, the I/O to produce the bigger data set is a lot lower than when using a DATA step. The advantages of using a SET statement in a DATA step include the ability to rename the variables from each data set, change lengths or other attributes, or manipulate the data.

**Example Code:**
```
data allsale;
   set prdsal2(rename=(predict=orig_predict)) sale2000(rename=(predict=orig_predict));
   predict=orig_predict*1.01;
run;
```

**COMPARE PROCEDURE**

**Description:** PROC COMPARE compares the variable values and attributes of two data sets and generates a report of the differences.

**Challenges:**
- The reports that are generated by PROC COMPARE contain the differences between the two data sets. It does not display values that are the same.
- When specified, the COMPARE procedure compares the data sets according to the BY and ID values. However, it does not provide a simple list of BY or ID values that are in one data set but not the other, or that are in both.
- There are no options with PROC COMPARE that will prevent the numeric values from being displayed in scientific notation or character values from being truncated at 20 characters. PROC COMPARE does not have an underlying table template that can be modified with PROC TEMPLATE to change the formatting of the printed output. An output data set must be created and then printed to control the formatting for long values.
- It is difficult to look among all the differences for many records and variables that could all be different.
**Best Use Case:** Use PROC COMPARE when you are confident that the data sets are very nearly the same and you want to find the one or two variables that might have differences.

When developing the program or when expecting a large number of differences, it might be best to suppress the output generated by PROC COMPARE by using the NOPRINT option. Instead, create a data set and then print that.

**Example Code:**

```sas
proc compare data=sashelp.prdsal2 comp=allsale noprint outdif outcomp outnoequal outbase out=diffs;
run;
```

**Comparisons:** There is no other procedure that has similar functionality to PROC COMPARE. It is possible to merge data sets and use assignment logic to compare the values of two data sets. That would be a substantial programming challenge, based on the number of variables and their type along with the desired structure of the final data set.

Alternatively, in theory, you can compare the two files by converting the data sets to text files and then using a comparison tool that is not offered by SAS.

**CONTENTS PROCEDURE**

**Description:** PROC CONTENTS provides data set attribute information and engine and host information, as well as a list of variables within the data set. The attribute information includes the number of observations, the number of variables, whether a data set is sorted, whether it has indexes, and what its encoding is. The engine and host information section provides the file size and the version of SAS and the operating system that created the data set. For the list of variables, PROC CONTENTS provides details about types, lengths, formats, informats, and labels.

PROC CONTENTS does not read or process the data. It reads only the data set header information.

**Challenges:**

- PROC CONTENTS does not report the number of observations for a DATA step view.
- The Created by and Last Modified by characteristics are unique to z/OS sequential access bound libraries. PROC CONTENTS does not provide these characteristics in UNIX or Microsoft Windows environments.
- PROC CONTENTS does not provide the number of observations that are contained in an external database, such as an Oracle table.

**Best Use Case:** Use PROC CONTENTS to check that your data set has the variables that you expect it to have, including their labels, types, and formats (Output 1 shows an example of a list of variables). The generated report is also a good way to see the SAS version and the encoding that was used to create the data set, which is critical information if you get a cross-environment data access error after you try to use the data set.

**Example Code:**

```sas
proc contents data=sashelp.prdsal2 out=contents;
run;
```
Output 1. Variable List Created with PROC CONTENTS

**Comparisons:** A PROC SQL step that references a DICTIONARY table (columns, indexes, members, and tables) can provide much of the same information that PROC CONTENTS does.

DICTIONARY.INDEXES provides more information about indexes than PROC CONTENTS. The below code demonstrates the PROC SQL step.

**Example Code:**

```sql
ods output indexes=indexzip1;
proc contents data=sashelp.zipcode out=contentszip;
run;

proc sql;
   create table indexzip2
       as select *
       from dictionary.indexes
       where libname='SASHELP' and memname='ZIPCODE';
quit;
```

**DATASETS PROCEDURE**

**Description:** PROC DATASETS enables you to change variable attributes, rename variables, create indexes, and see the contents of a data set.

**Challenges:**

- PROC DATASETS enables you to copy and rename variables only. It cannot drop variables.
- It does not provide the number of observations that are contained in an external database table.

**Best Use Case:**

- PROC DATASETS is often used to delete all the data sets in a library. However, be careful when you do this, because once deleted, you cannot get them back.
Example Code:

```sas
proc datasets library=work kill;
quit;
```

- Use PROC DATASETS to modify formats and labels without having to run a DATA step—it is less resource-intensive (Output 2 shows a log that confirms that modification is successful).

Example Code:

```sas
data work.prdsal2;
  set sashelp.prdsal2;
run;
```

```sas
proc datasets library=work nolist;
  modify prdsal2;
  label product='Product Name';
  format country $char20.;
quit;
```

```
5
6 /****Best Use Case****/
7 /*Modify labels and formats*/
8 proc datasets library=work nolist;
NOTE: Writing HTML Body file: sashtml.htm
9 modify prdsal2;
10 label product='Product Name';
11 format country $char20.;
12 quit;

NOTE: MODIFY was successful for WORK.PRDSAL2.DATA.
NOTE: PROCEDURE DATASETS used (Total process time):
    real time          0.20 seconds
    cpu time           0.12 seconds
```

Output 2. Log Showing That a PROC DATASETS Modification Is Successful

Comparisons:

- PROC DATASETS and PROC CONTENTS provide the same information about either a data set or all the data sets in a library. For content information, use the procedure that you are most comfortable with.

Example Code:

```sas
proc datasets library=work;
  contents data=prdsal2;
quit;
```

```sas
proc contents data=prdsal2;
run;
```
PROC SQL provides data set information and enables you to change a variable name at the same time. Keep in mind that PROC SQL will process the entire data set, not just the header information.

Example Code:

```sql
proc datasets library=work nolist;
modify prdsal2;
index create state / nomiss;
format monyr yymon7.;
quit;

proc sql noprint;
create table prdsal2b as
select country, state, county, actual, predict, prodtype, product,
 year, quarter as qtr, month, monyr format=yymon7.
from sashelp.prdsal2;
create index state on prdsal2b(state);
quit;
```

FORMAT PROCEDURE

**Description:** PROC FORMAT creates informats, which are instructions for how Base SAS should read and interpret raw values. It also creates formats and picture formats, which are instructions for how to display data values.

Formats are created either by hardcoding the values or by using a CNTLIN= data set, which contains variables with the desired value-label combinations.

**Challenges:** Forethought is necessary: you must know how you want your data values displayed.

**Best Use Case:**
- Generally, PROC FORMAT is used to create a format so that raw values, such as zero and one, are displayed as another value, such as No and Yes.

Example Code:

```sas
data work.prdsal2;
  set sashelp.prdsal2;
run;

proc format;
  value $ typ
    'FURNITURE'='Residential'
    'OFFICE'='Commercial'
    other='Other';
run;

data work.prdsal2;
  set sashelp.prdsal2;
run;

proc format;
  value $ typ
    'FURNITURE'='Residential'
    'OFFICE'='Commercial'
    other='Other';
run;

(code continued)
```
Creating a format is also needed when you want to reassign values (in a DATA step); to group values or categories when calculating statistics or creating reports in other procedures; or to generate rows or placeholders for categories that are not present in the data. The code below demonstrates using a format to group values in PROC FREQ output.

**Example Code:**
```
proc format;
  value $ func
    'BED', 'SOFA'='Soft'
    'DESK','CHAIR'='Hard';
run;

proc freq data=sashelp.prdsal2;
  format product $func.;
  table product;
run;
```

**Comparisons:** There is no other procedure that has similar functionality to PROC FORMAT. In a DATA step, you can choose to use IF-THEN or IF-THEN/ELSE statements to conditionally assign a new value based on an old value. To an extent, one method is not better than the other, because you still must type out both the new and old values.

**Example Code:**
```
data work.prdsal2;
  set sashelp.prdsal2;
run;

data prdsal2b;
  length prodtype $12;
  set prdsal2;
  if prodtype='FURNITURE' then prodtype='Residential';
  else if prodtype='OFFICE' then prodtype='Commercial';
  else prodtype='Other';
run;
```

**SORT PROCEDURE**

**Description:** The SORT procedure sorts data and can remove duplicate observations.

**Challenges:** Sorting data is a common, necessary practice. However, PROC SORT will create a temporary utility file that is approximately twice the size of the original data set. The amount of memory and disk space used by PROC SORT can be problematic for large data sets.

**Best Use Case:** You should sort your data with PROC SORT when the other procedures that you are using for analytics and reporting require sorting. However, data that is sorted or grouped prior to being brought into SAS does not need re-sorting. In most procedures, you can take advantage of the NOTSORTED option in the BY statement. This option will group together observations that have the same BY value. However, it does not put them into any specific order, such as by numeric value. Also, using a CLASS statement within a procedure does not require the data to be presorted.
Example Code:

```sas
proc sort data=sashelp.prdsal2 out=prdsal2;
  by year product country;
run;
```

Comparisons: PROC SQL can sort data sets. The code below demonstrates using both PROC SORT and PROC SQL to sort a data set and keep only the unique combinations of COUNTRY, STATE, and PRODUCT. The two steps generate identical data sets.

Example Code:

```sas
proc sort data=sashelp.prdsal2 out=unqprd1(keep=country state product) nodupkey;
  by country state product;
run;
```

```sas
proc sql noprint;
  create table unqprd2 as
  select distinct country, state, product
  from sashelp.prdsal2
  order by country, state, product;
quit;
```

SQL PROCEDURE

Description: PROC SQL is kind of a super utility procedure. It lets you sort data, merge data sets, create indexes, and rename variables. You can also create macro variables and calculate statistics. The procedure accesses DICTIONARY tables, which are special read-only PROC SQL tables or views that contain information about all the SAS libraries and SAS data sets.

Challenges:

- It is not a standard SAS procedure that requires a certain set of statements or options. There is a learning curve to understand the syntax and take full advantages of its capabilities.
- It does not create a production-quality report, and you cannot manipulate the printed table with style overrides.

Best Use Case:

- For the average SAS programmer, PROC SQL is best used for generating macro variables in one step, such as a list of unique variable values, population totals, or minimum and maximum values.

Example Code:

```sas
proc sql noprint;
  select count(country) into :usa from sashelp.prdsal2
  where country="U.S.A.";
  select count(country) into :mex from sashelp.prdsal2
  where country="Mexico";
  select count(country) into :can from sashelp.prdsal2
  where country="Canada";
quit;
```

```sas
proc sql noprint;
  select distinct(state) into :stlist1 separated by ', ' from sashelp.prdsal2;
quit;
```
• Access to DICTIONARY tables is also very useful when you need information about libraries, data sets, and variables.

**Example Code:**

```sql
proc sql;
  select * 
  from dictionary.tables
  where libname='WORK';
quit;
```

**Comparisons:** PROC SQL functionality overlaps that of the DATASETS, CONTENTS, FREQ, MEANS, and SORT procedures. It also has some of the same capabilities as the DATA step.

It is difficult to provide a set rule for when to use it over other procedures. If you need a Cartesian product, use PROC SQL. If you understand merging with the IN= operator more than joins, use a DATA step with the MERGE statement.

The code below demonstrates generating the same counts with both PROC SQL and PROC FREQ.

**Example Code:**

```sql
proc sql;
  select product, count(product) as n 
  from sashelp.prdsal2
  group by product
  order by product;
quit;
```

```sql
proc freq data=sashelp.prdsal2;
  tables product /nopercent nocum;
run;
```

**TRANSPOSE PROCEDURE**

**Description:** PROC TRANSPOSE restructures data. It changes data sets from long and narrow to wide and short, or vice versa.

**Challenges:**

• It usually requires that data be sorted first.

• Controlling the resulting variable names can be challenging; to do that, sometimes you must use two PROC TRANSPOSE steps.

**Best Use Case:** Use PROC TRANSPOSE with a data set that is unique—either within a set of BY variables or over the entire data set—where you need to create new numeric variables named after the value of a categorical variable.

**Example Code:**

```sql
proc transpose data=sashelp.stocks out=test;
  where year(date)=2005;
  by stock;
  id date;
  var close;
run;
```

**Comparisons:** Using a DATA step can transpose data with the use of ARRAY, DO, and OUTPUT statements. It often provides more control over the resulting variables, and you can include any other DATA step functionality that you need. Using a DATA step as an alternative to PROC TRANSPOSE is daunting for some programmers because of the use of arrays.
Transposing multiple variables is a challenging task. The code below provides a comparison of the steps needed to use PROC TRANSPOSE and a DATA step that accomplishes the same task.

Transposing multiple variables with PROC TRANSPOSE requires multiple steps: a PROC SORT step, two PROC TRANSPOSE steps, and a DATA step.

Example Code:

```sas
proc sort data=sashelp.prdsale out=prdsale;
  where product='SOFA' and year=1993 and region='EAST';
  by country division prodtype;
run;

proc transpose data=prdsale out=sale1 prefix=predict_
  by country division prodtype;
  id month;
  var predict;
run;

proc transpose data=prdsale out=sale2 prefix=actual_
  by country division prodtype;
  id month;
  var actual;
run;

data all1;
  merge sale1 sale2;
  by country division prodtype;
run;
```

The DATA step code below accomplishes the same task of transposing the data as the previous four steps.

Example Code:

```sas
data all2;
  set prdsale;
  by country division prodtype;
  array pred{12} predict_Jan predict_Feb predict_Mar predict_Apr
       predict_May predict_Jun predict_Jul predict_Aug
       predict_Sep predict_Oct predict_Nov predict_Dec;
  array actl{12} actual_Jan actual_Feb actual_Mar actual_Apr
       actual_May actual_Jun actual_Jul actual_Aug actual_Sep
       actual_Oct actual_Nov actual_Dec;
  array vals{12} $ ('Jan' 'Feb' 'Mar' 'Apr' 'May' 'Jun' 'Jul' 'Aug'
       'Sep' 'Oct' 'Nov' 'Dec');
  retain predict: actual:;

  do i=1 to 12;
    if put(month,monname3.)=vals{i} then do;
      pred{i}=predict;
      actl{i}=actual;
    end;
  end;
  if last.prodtype;
  drop actual predict quarter year vals: i;
run;
```
STATISTICAL-ESQUE PROCEDURES

For the purposes of this paper, statistical-esque procedures are those that enable you to summarize your data and calculate various statistics. The results might or might not be in final, report-quality form.

FREQ PROCEDURE

Description: PROC FREQ gives counts and percentages. It can also generate statistics, like Fisher’s exact test or the chi-square test.

Challenges:
- PROC FREQ will not insert categories that are not present in the data with a count of zero, even when you apply a format that contains the category that does not exist in the data.
- The OUT= and ODS output data sets do not contain the same variables or structure.
- It is not easy to change the format of the statistics for one-way tables. For example, you cannot display the counts with commas without making a change to the underlying table template.
- You can get overall percentages for a whole data set or within a BY group, but you cannot generate customized percentages.

Best Use Case:
- PROC FREQ is often used to create data sets with total counts that can be used later as denominators, when merged with detail data.
- PROC FREQ is also extremely useful for checking your data to ensure that it contains all expected categories. An example is shown in Output 3.

Example Code:

```
proc freq data=sashelp.prdsal2;
  tables country*product / list out=freqs;
run;
```
### The FREQ Procedure

<table>
<thead>
<tr>
<th>COUNTRY</th>
<th>PRODUCT</th>
<th>Frequency</th>
<th>Percent</th>
<th>Cumulative Frequency</th>
<th>Cumulative Percent</th>
</tr>
</thead>
<tbody>
<tr>
<td>Canada</td>
<td>BED</td>
<td>1152</td>
<td>5.00</td>
<td>1152</td>
<td>5.00</td>
</tr>
<tr>
<td>Canada</td>
<td>CHAIR</td>
<td>1152</td>
<td>5.00</td>
<td>2304</td>
<td>10.00</td>
</tr>
<tr>
<td>Canada</td>
<td>DESK</td>
<td>1152</td>
<td>5.00</td>
<td>3456</td>
<td>15.00</td>
</tr>
<tr>
<td>Canada</td>
<td>SOFA</td>
<td>1152</td>
<td>5.00</td>
<td>4608</td>
<td>20.00</td>
</tr>
<tr>
<td>Mexico</td>
<td>BED</td>
<td>1152</td>
<td>5.00</td>
<td>5760</td>
<td>25.00</td>
</tr>
<tr>
<td>Mexico</td>
<td>CHAIR</td>
<td>1152</td>
<td>5.00</td>
<td>6912</td>
<td>30.00</td>
</tr>
<tr>
<td>Mexico</td>
<td>DESK</td>
<td>1152</td>
<td>5.00</td>
<td>8064</td>
<td>35.00</td>
</tr>
<tr>
<td>Mexico</td>
<td>SOFA</td>
<td>1152</td>
<td>5.00</td>
<td>9216</td>
<td>40.00</td>
</tr>
<tr>
<td>U.S.A.</td>
<td>BED</td>
<td>3456</td>
<td>15.00</td>
<td>12672</td>
<td>55.00</td>
</tr>
<tr>
<td>U.S.A.</td>
<td>CHAIR</td>
<td>3456</td>
<td>15.00</td>
<td>16128</td>
<td>70.00</td>
</tr>
<tr>
<td>U.S.A.</td>
<td>DESK</td>
<td>3456</td>
<td>15.00</td>
<td>19584</td>
<td>85.00</td>
</tr>
<tr>
<td>U.S.A.</td>
<td>SOFA</td>
<td>3456</td>
<td>15.00</td>
<td>23040</td>
<td>100.00</td>
</tr>
</tbody>
</table>

#### Output 3. PROC FREQ Output

**Comparisons:** PROC REPORT and PROC TABULATE also calculate counts and percentages. They might require more coding to get the same information, but they enable you to have more control over formatting and table structure.

The PRELOADFMT option, in both the REPORT and TABULATE procedures, will display values that are present in the format but are not present in the data set.

PROC TABULATE can calculate subtotals, which PROC FREQ cannot.

**Example Code:**

```sas
proc tabulate data=sashelp.prdsal2 format=8.;
  class country product;
  table country*(product all='SubTotal') all='Total', N;
run;
```

### MEANS AND SUMMARY PROCEDURES

**Description:** PROC MEANS calculates statistics, such as N, sum, mean, minimum, maximum, standard error, and percentiles. PROC SUMMARY is very similar to PROC MEANS: it computes the same descriptive statistics. By default, the SUMMARY procedure does not produce printed output. Also, if the VAR statement is omitted, PROC SUMMARY produces a simple count of observations.

**Challenges:**

- The procedures do not calculate percentages.
- The structure of the report output is either difficult or impossible to change.
**Best Use Case:** Use PROC MEANS when you want to generate data sets with statistics, totals, and overall. Simultaneously, use it to add records, with a count of zero, for values that are not present in the data. See Output 4 for an example of the PROC MEANS output.

**Example Code:**

```plaintext
proc format;
   value $prod 'BED'='BED' 'CHAIR'='CHAIR'
      'DESK'='DESK' 'SOFA'='SOFA' 'TABLE'='TABLE';
run;

proc means data=sashelp.prdsal2 completetypes nway;
   format product $prod.;
   class product / preloadfmt;
   var actual predict;
   output out=means mean= sum= n= /autoname;
run;
```

### The MEANS Procedure

<table>
<thead>
<tr>
<th>Product</th>
<th>N Obs</th>
<th>Variable</th>
<th>Label</th>
<th>N</th>
<th>Mean</th>
<th>Std Dev</th>
<th>Minimum</th>
<th>Maximum</th>
</tr>
</thead>
<tbody>
<tr>
<td>BED</td>
<td>5760</td>
<td>ACTUAL</td>
<td>Actual Sales Predicted Sales</td>
<td>5760</td>
<td>644.14909447</td>
<td>654.53234735</td>
<td>0.0615385</td>
<td>672.2739927</td>
</tr>
<tr>
<td>CHAIR</td>
<td>5760</td>
<td>ACTUAL</td>
<td>Actual Sales Predicted Sales</td>
<td>5760</td>
<td>641.5596563</td>
<td>640.5110277</td>
<td>0</td>
<td>3376.20</td>
</tr>
<tr>
<td>DESK</td>
<td>5760</td>
<td>ACTUAL</td>
<td>Actual Sales Predicted Sales</td>
<td>5760</td>
<td>645.5484375</td>
<td>656.2886439</td>
<td>0</td>
<td>3415.30</td>
</tr>
<tr>
<td>SOFA</td>
<td>5760</td>
<td>ACTUAL</td>
<td>Actual Sales Predicted Sales</td>
<td>5760</td>
<td>674.3301707</td>
<td>673.9000104</td>
<td>0</td>
<td>3515.60</td>
</tr>
<tr>
<td>TABLE</td>
<td>0</td>
<td>ACTUAL</td>
<td>Actual Sales Predicted Sales</td>
<td>0</td>
<td>-</td>
<td>-</td>
<td>-</td>
<td>-</td>
</tr>
</tbody>
</table>

**Output 4. PROC MEANS Output**

### Comparisons:
- PROC TABULATE calculates the same statistics as the MEANS and SUMMARY procedures, as well as percentages.

**Example Code:**

```plaintext
proc tabulate data=sashelp.prdsal2;
   class year country;
   var actual;
   table country='', year*actual=''*(mean median);
run;
```
• PROC SQL calculates statistics and has the additional capability of generating macro variables with the calculated values.

Example Code:

```sql
proc sql noprint;
  select max(actual) into :actmax from sashelp.prdsal2;
quit;
```

UNIVARIATE PROCEDURE

Description: PROC UNIVARIATE calculates summary statistics, generates quantiles and percentiles, performs tests for goodness of fit, and produces probability plots.

Challenges:

• The default output contains a large number of tables and data. To limit the generated output, you have to use an ODS SELECT statement or an ODS EXCLUDE statement.

• There are no options in the PROC UNIVARIATE statement for subsetting to get only the statistics that you are interested in. You have to create an output data set and manipulate it to get only the statistics that you need.

Best Use Case: Use it to generate custom (nondefault) percentiles.

Example Code:

```sas
proc univariate data=sashelp.prdsal2;
var actual;
output out=univ pctlpts=15 25 35 pctlpre=actual_p;
run;
```

Comparisons: PROC MEANS calculates the same summary statistics as PROC UNIVARIATE. If you do not need the statistical testing or plotting capabilities, either procedure can be used. It is a matter of programmer preference.

Example Code:

```sas
proc means data=sashelp.prdsal2 min max mean mode;
  class product;
  var predict actual;
  output out=stats mean= sum= /autoname;
run;
```

REPORTING PROCEDURES

For the purposes of this paper, reporting procedures are those that are used to create final, publication-quality report tables.

PRINT PROCEDURE

Description: The PRINT procedure displays data. Summary totals can be requested for numeric variables.

Challenges:

• The amount of customization in the final table is limited. For example, PROC PRINT cannot either apply formatting based on another column or include additional text.

• The values of numeric columns are simply displayed. Their values cannot be added, subtracted, multiplied, or divided against each other.

• An output data set cannot be created. There is no OUT= data set option, and PROC PRINT does not support the ODS OUTPUT statement.
Best Use Case:

- Use PROC PRINT when you need a quick view of your data to ensure it has the variables and values that you expect.

Example Code:

```sas
proc print data=sashelp.prdsal2(obs=30);
   var country state product actual predict;
run;
```

- PROC PRINT is the easiest way to get a report when your data set contains everything you need and it does not require additional formatting, or you have a small amount of data and want a grand total. Output 5 provides an example with a summary row at the end.

Example Code:

```sas
proc print data=sashelp.class grandtotal_label='Grand Total';
   var name age height weight;
   sum height weight;
run;
```

<table>
<thead>
<tr>
<th>Obs</th>
<th>Name</th>
<th>Age</th>
<th>Height</th>
<th>Weight</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Alfred</td>
<td>14</td>
<td>69.0</td>
<td>112.5</td>
</tr>
<tr>
<td>2</td>
<td>Alice</td>
<td>13</td>
<td>56.5</td>
<td>84.0</td>
</tr>
<tr>
<td>3</td>
<td>Barbara</td>
<td>13</td>
<td>66.3</td>
<td>98.0</td>
</tr>
<tr>
<td>4</td>
<td>Carol</td>
<td>14</td>
<td>62.8</td>
<td>102.5</td>
</tr>
<tr>
<td>5</td>
<td>Henry</td>
<td>14</td>
<td>63.5</td>
<td>102.5</td>
</tr>
<tr>
<td>6</td>
<td>James</td>
<td>12</td>
<td>57.3</td>
<td>83.0</td>
</tr>
<tr>
<td>7</td>
<td>Jane</td>
<td>12</td>
<td>59.8</td>
<td>84.5</td>
</tr>
<tr>
<td>8</td>
<td>Janet</td>
<td>15</td>
<td>62.5</td>
<td>112.5</td>
</tr>
<tr>
<td>9</td>
<td>Jeffrey</td>
<td>13</td>
<td>62.5</td>
<td>84.0</td>
</tr>
<tr>
<td>10</td>
<td>John</td>
<td>12</td>
<td>59.0</td>
<td>99.5</td>
</tr>
<tr>
<td>11</td>
<td>Joyce</td>
<td>11</td>
<td>51.3</td>
<td>50.5</td>
</tr>
<tr>
<td>12</td>
<td>Judy</td>
<td>14</td>
<td>64.3</td>
<td>90.0</td>
</tr>
<tr>
<td>13</td>
<td>Louise</td>
<td>12</td>
<td>56.3</td>
<td>77.0</td>
</tr>
<tr>
<td>14</td>
<td>Mary</td>
<td>15</td>
<td>66.6</td>
<td>112.0</td>
</tr>
<tr>
<td>15</td>
<td>Philip</td>
<td>16</td>
<td>72.0</td>
<td>150.0</td>
</tr>
<tr>
<td>16</td>
<td>Robert</td>
<td>12</td>
<td>64.8</td>
<td>128.0</td>
</tr>
<tr>
<td>17</td>
<td>Ronald</td>
<td>15</td>
<td>67.0</td>
<td>133.0</td>
</tr>
<tr>
<td>18</td>
<td>Thomas</td>
<td>11</td>
<td>57.5</td>
<td>85.0</td>
</tr>
<tr>
<td>19</td>
<td>William</td>
<td>15</td>
<td>66.5</td>
<td>112.0</td>
</tr>
<tr>
<td></td>
<td>Grand Total</td>
<td></td>
<td>1184.4</td>
<td>1900.5</td>
</tr>
</tbody>
</table>

Output 5. PROC PRINT Output with a Summary Row
Comparisons: PROC REPORT can simply display data, but it can also apply custom formatting and transpose the data.

Example Code:

```sas
proc report data=sashelp.prdsal2;
  column country product predict actual;
  define country / group;
  define product / group;
  break after country / summarize;
  compute actual;
    if actual.sum > predict.sum then call define('actual.sum','style', 'style=[foreground=green]');
  endcomp;
run;
```

REPORT PROCEDURE

Description: PROC REPORT generates both detailed and summary reports, meaning it can print all observations from your data set or group them into categories. It can calculate statistics, create new variables, and transpose data.

Challenges:

- There is a learning curve to using the compute block, the most powerful aspect of PROC REPORT.
- It does not provide a subtotals breakdown like PROC TABULATE can generate.

Best Use Case: Use it to create a final report that includes customized headers and text, coloring based on certain values (trafficlighting), and calculations based on various columns. See Output 6 for an example of a report that uses a customized header.

Example Code:

```sas
proc report data=sashelp.prdsal2;
  column country state product predict actual diff;
  define country / group noprint;
  define state / group;
  define product / group;
  define diff / computed 'Dif=Pred-Act' format=dollar10.2;
  break after country / summarize;

  compute before country;
    line 'Section for ' country $20.;
  endcomp;
  compute after country;
    state='Total';
  endcomp;
  compute diff;
    diff=predict.sum-actual.sum;
  endcomp;
run;
```
Comparisons: PROC REPORT combines a lot of the capabilities of the PRINT and TABULATE procedures. PROC PRINT should be used for small, simple reports. PROC TABULATE easily generates subtotals and percentages that require a large amount of code in PROC REPORT.

Example Code:

```plaintext
proc format;
    picture mypct low-high='009.9%';
run;

proc tabulate data=sashelp.prdsal2;
    class prodtype product;
    var actual;
    table prodtype*product, actual*(sum*f=dollar12.
        pctsum<product>*f=mypct.);
run;
```
TABULATE PROCEDURE

Description: The TABULATE procedure calculates statistics and percentages. It is capable of summarizing at a categorical level and generating overall totals. It can create columns from values in the input data set.

Challenges:

- The syntax for the TABLE statement has a learning curve. The statement can be long, and it might be confusing due to the asterisks, brackets, and denominator definitions.
- The procedure cannot perform basic mathematical calculations, such as adding, subtracting, multiplying, or dividing analysis variables. However, there is a special case where you can calculate a ratio by using two analysis variables and the PCTSUM statistic.
- You can apply formatting to a variable, but that formatting cannot be based on the value of another column.
- You cannot insert text in the middle of the table.

Best Use Case: PROC TABULATE generates percentages with very little code. It can also calculate totals and subtotals for various combinations of class variables. Output 7 shows an example.

Example Code:

```plaintext
proc tabulate data=sashelp.prdsal2;
  class country product;
  var actual;
  tables country*product all='TOTALS'*product='',
             actual*(sum*f=dollar12. pctsum<product>);
run;
```
Output 7. PROC TABULATE Output

**Comparisons:** PROC REPORT generates the same statistics as PROC TABULATE. Its additional capabilities are manipulating cell values and formatting based on other columns.

**Example Code:**

```sas
proc report data=sashelp.prdsal2;
  column country product predict actual diff;
  define country / group;
  define product / group;
  define diff / computed 'Dif=Pred-Act' format=dollar10.2;
  break after country / summarize;
  compute diff;
    diff=predict.sum-actual.sum;
    if _break_ ^= '' then do;
      if country='U.S.A.' then call define('diff','style', 'style=[background=lightblue]');
      else if country='Mexico' then call define('diff','style', 'style=[background=lightgreen]');
      else if country='Canada' then call define('diff','style', 'style=[background=lightred]');
    end;
  endcomp;
run;
```
CONCLUSION
This paper focused on commonly used Base SAS procedures. The capability of most procedures overlaps with at least one other procedure. Each procedure has its strengths and its challenges. Use the information in this paper to help you decide whether it would be easier to accomplish a task with a different procedure. Or you can confirm that you are already doing the right thing!

REFERENCES


ACKNOWLEDGMENTS
The author is immensely grateful to Marcia Surratt, Kathryn McLawhorn, Sharon Hamrick, and Jerry Leonard for contributing to this paper, and to Chris Dahlin, who edited the paper.

RECOMMENDED READING


CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:
Jane Eslinger
SAS Institute Inc.
SAS Campus Drive
Cary, NC 27513
Email: support@sas.com
Web: support.sas.com/en/support-home.html